# Maintainer: Jonathan Steel <jsteel@aur.archlinux.org>
# Contributor: Justin Davis (juster) <jrcd83@gmail.com>

pkgname=perl-sub-install
pkgver=0.928
pkgrel=9
pkgdesc="Install subroutines into packages easily"
arch=('any')
url="https://metacpan.org/release/Sub-Install"
license=('PerlArtistic' 'GPL')
depends=('perl')
options=('!emptydirs')
source=(https://search.cpan.org/CPAN/authors/id/R/RJ/RJBS/Sub-Install-$pkgver.tar.gz)
sha256sums=('61e567a7679588887b7b86d427bc476ea6d77fffe7e0d17d640f89007d98ef0f')

build() {
  cd Sub-Install-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd Sub-Install-$pkgver
  make test
}

package() {
  cd Sub-Install-$pkgver
  make DESTDIR="$pkgdir" install
}
